
               0900 *APPLE IIE ROM PATCH FOR BREAKING INTO DEBUGGER
               0950 *PART 2: THIS ROUTINE REPLACES THE SELF-TEST
               0960 *ROM CODE AND STARTS AT $C600.
               0961 *WHEN INVOKED THE USER CAN
               0962 *PRESS EITHER ESC OR SPC.
               0963 *ESC-GO DIRECTLY TO MONITOR
               0964 *SPC-COPY MEMORY LOCATIONS $0000-$08FF
               0965 *    TO $2000-$2FFF THEN JUMP TO MONITOR
               0966 * ANY OTHER KEY - PROCEED NORMALLY
               0970 *CALLED FROM $FECD WHEN RESET IS PRESSED
               0980 *THANKS TO THE COMPUTIST MAGAZINE (ISSUE #6)
               0981 *AND KRAKOWITZ TEXT FILES FOR THE INSPIRATION
003C-          0985 A1L     .EQ $003C    LSB ADDRESS
003D-          0986 A1H     .EQ $003D    MSB ADDRESS
003E-          0987 A2L     .EQ $003E    LSB OF ADDRESS 2
003F-          0988 A2H     .EQ $003F    MSB OF ADDRESS 2
0042-          0989 A4L     .EQ $0042    LSB OF ADDRESS 4
0043-          0990 A4H     .EQ $0043    MSB OF ADDRESS 4
FE2C-          0991 MONMOVE .EQ $FE2C    MONITOR MOVE ROUTINE
FC58-          0992 HME     .EQ $FC58    CLEARS TEXT SCREEN
FB2F-          0993 TXT     .EQ $FB2F    SELECTS TEXT SCREEN
FECD-          1001 MONEXIT .EQ $FECD   PART 3 AND TAPE WRITE SECTION
C000-          1002 KEYBD  .EQ $C000    KEYBOARD
FF03-          1003 NORMAL .EQ $FF03    EXIT LOCATION IN PART 1
               1005        .OR $C600    LOCATE IN SELF-TEST SECTION OF MEM
               1007        .TA $800
C600- AD 00 C0 1010 START  LDA KEYBD    LOOK AT KEYBOARD
C603- F0 FB    1020        BEQ START    KEEP READING UNTIL A KEY PRESSED
C605- C9 A0    1025        CMP #$A0     WAS SPC PRESSED?
C607- F0 07    1030        BEQ CPY      YES, START MEM COPY
C609- C9 9B    1035        CMP #$9B     WAS ESC PRESSED?
C60B- F0 3C    1037        BEQ END      YES, JUMP STRAIGHT TO MONITOR
C60D- 4C 03 FF 1040        JMP NORMAL   BACK TO PART 1 FOR PASSTHROUGH
C610- 8D 02 29 1042 CPY    STA $2902    SAVE REGISTERS
C613- 8E 03 29 1043        STX $2903
C616- 8C 04 29 1044        STY $2904
C619- BA       1050        TSX
C61A- 8E 01 29 1060        STX $2901    
C61D- A0 00    1062        LDY #0       START WITH BYTE 0
C61F- B9 00 00 1064 SAVE1  LDA 0,Y      GET A ZERO PAGE BYTE
C622- 99 00 20 1066        STA $2000,Y  SAVE IT
C625- B9 00 01 1090        LDA $100,Y   GET A STACK BYTE
C628- 99 00 21 1100        STA $2100,Y  SAVE IT
C62B- C8       1110        INY          
C62C- D0 F1    1120        BNE SAVE1    IF NOT DONE, MOVE ANOTHER BYTE
C62E- 84 3C    1125        STY A1L      MSB FROM ADDRESS IS 0
C630- 84 42    1130        STY A4L      MSB OF THE TO ADDRESS IS 0
C632- 84 3E    1135        STY A2L      MOVE CODE THROUGH $900
C634- A9 09    1140        LDA #9       
C636- 85 3F    1145        STA A2H      
C638- A9 02    1150        LDA #2       MOVE CODE STARTING AT $200
C63A- 85 3D    1155        STA A1H      
C63C- A9 22    1160        LDA #$22     MOVE CODE INTO PAGE 22
C63E- 85 43    1165        STA A4H
C640- 20 2C FE 1170        JSR MONMOVE
C643- 20 2F FB 1175        JSR TXT
C646- 20 58 FC 1180        JSR HME
C649- 4C CD FE 1220 END    JMP MONEXIT  JUMP TO PART 3 AND EXIT TO MON
               1230        .EN

SYMBOL TABLE

003D- A1H



003D- A1H
003C- A1L
003F- A2H
003E- A2L
0043- A4H
0042- A4L
C610- CPY
C649- END
FC58- HME
C000- KEYBD
FECD- MONEXIT
FE2C- MONMOVE
FF03- NORMAL
C61F- SAVE1
C600- START
FB2F- TXT

